

August 2024 – file containing SPARQL scripts that can be run on the
cancerIndicatorOntologyExample ontology to demonstrate the SOLICIT framework.
All the examples in this file can be run individually using the SPARQL Query tab in the Protégé
ontology-editor user interface, apart from example #7 that requires a SPARQL endpoint on a
remote server.
The SPARQL queries address how the SOLICIT framework can tackle the contextualization
requirements identified by the good indicators guide: Ref - British National Health Service (NHS)
Institute for Innovation and Improvement. (2024). The good indicators guide: Understanding how
to use and choose indicators. [https://www.england.nhs.uk/improvement-hub/wp-](https://www.england.nhs.uk/improvement-hub/wp-content/uploads/sites/44/2017/11/The-Good-Indicators-Guide.pdf)
content/uploads/sites/44/2017/11/The-Good-Indicators-Guide.pdf

PREFIX rdf: <<http://www.w3.org/1999/02/22-rdf-syntax-ns#>>
PREFIX owl: <<http://www.w3.org/2002/07/owl#>>
PREFIX rdfs: <<http://www.w3.org/2000/01/rdf-schema#>>
PREFIX xsd: <<http://www.w3.org/2001/XMLSchema#>>
PREFIX obo: <<http://purl.obolibrary.org/obo/>>
PREFIX cco: <<http://www.ontologyrepository.com/CommonCoreOntologies/>>
PREFIX slcto: <https://data.jrc.ec.europa.eu/dataset/SOLICIT_IndicatorOntology#>
PREFIX ncdo: <<https://data.jrc.ec.europa.eu/dataset/SOLICIT/NCDIndicatorOntology#>>
PREFIX cio: <https://data.jrc.ec.europa.eu/dataset/SOLICIT/CancerIndicatorOntology#>

1) what is being measured? - complementarity of indicator attributes

```
SELECT DISTINCT *  
  WHERE { cio:AStRIndicatorCaEU2010Ex ?predicate ?object .  
          FILTER (?object != owl:NamedIndividual) .  
          ?object rdf:type ?parent_class .  
          FILTER (?parent_class != owl:NamedIndividual) .  
        } ORDER BY DESC( ?object)
```

2) why is it being measured? - scope

```
SELECT DISTINCT ?scope  
  WHERE { cio:AStRIndicatorCaEU2010Ex ?predicate ?object .  
          ?object rdf:type ?parent_class .  
          ?parent_class ?p21 ?o1 .  
          ?o1 owl:someValuesFrom slcto:Indicator_OC_Scope .  
          ?object cco:is_tokenized_by ?scope  
        }  
}
```

3) how is the indicator actually defined?

```
SELECT DISTINCT ?definition  
  WHERE { cio:AStRIndicatorCaEU2010Ex ?predicate ?object .  
          ?object rdf:type ?parent_class .  
          ?parent_class ?p21 ?o1 .  
          ?o1 owl:someValuesFrom slcto:Indicator_OC_Definition .  
          ?object cco:is_tokenized_by ?definition  
        }  
}
```

4) who are the subjects of the measurement? - indicator location coverage

```
SELECT ?coverage_obj ?coverage_link
WHERE {
  cio:AStRIndicatorCaEU2010Ex ?pred ?ind_process .
  ?ind_process obo:BFO_0000200 ?coverage_obj .
  ?coverage_obj cco:is_tokenized_by ?coverage . BIND(str(?coverage) AS ?coverage_link) .
}

SELECT ?data_el ?obj_prop ?obj_prop_arg
WHERE {
  cio:AStRIndicatorCaEU2010Ex ?predicate ?object .
  ?object owl:someValuesFrom ?data_el .
  ?data_el ?subclass_of ?obj_prop_dec .
  FILTER EXISTS { ?obj_prop_dec owl:someValuesFrom
    ncdo:Geographical_Area_OC_Delimitation } .
  ?data_el ?subclass_of ?obj_prop . ?obj_prop owl:someValuesFrom ?obj_prop_arg .
}
```

5) what time period does the indicator cover?

```
SELECT ?time_period_obj ?time_period
WHERE {
  cio:AStRIndicatorCaEU2010Ex cco:describes ?ind_process .
  ?ind_process obo:BFO_0000199 ?time_period_obj .
  ?time_period_obj cco:is_tokenized_by ?time_period .
}

SELECT ?data_el ?obj_prop ?obj_prop_arg ?comment
WHERE {
  cio:AStRIndicatorCaEU2010Ex ?predicate ?object .
  ?object owl:someValuesFrom ?data_el .
  ?data_el ?subclass_of ?obj_prop_dec .
  FILTER EXISTS { ?obj_prop_dec owl:someValuesFrom
    ncdo:Time_OC_Delimitation } .
  ?data_el ?subclass_of ?obj_prop . ?obj_prop owl:someValuesFrom ?obj_prop_arg .
  OPTIONAL { ?obj_prop_arg rdfs:comment ?comment } .
}
```

6) what is the unit/dimension of the measurement?

```
SELECT ?obj_property ?component ?super_type ?is_tokenized_by
WHERE {
  cio:AStRIndicatorCaEU2010Ex cco:is_measured_by ?measured_by .
  ?measured_by ?obj_property ?component .
  FILTER (?obj_property != rdf:type) .
  ?component rdf:type ?super_type
  FILTER (?super_type != owl:NamedIndividual) .
  OPTIONAL { ?component cco:is_tokenized_by ?is_tokenized_by }
}
```

7) Where do the underlying data come from?

Access of the remote node to determine its associated attributes (e.g. local data file information).

N.B. this script requires a SPARQL endpoint on the server of the remote node

```
SELECT ?remote_obj_IRI ?pred ?obj
WHERE {
```

```

?predicate rdfs:label "continuant part of"@en .
cio:AStRIndicatorCaEU2010Ex ?predicate ?cont_part_of_1 .
?cont_part_of_1 rdf:type slcto:Indicator_OC_Provider_URI .
?cont_part_of_1 cco:is_tokenized_by ?remote_obj .
BIND (IRI(?remote_obj) AS ?remote_obj_IRI) .
?cont_part_of_1 obo:BFO_0000056 ?ext_process .
?ext_process cco:has_input ?input_obj .
?input_obj cco:is_tokenized_by ?SPARQL_endpoint .
BIND (IRI(?SPARQL_endpoint) AS ?endpoint_iri) .
SERVICE ?endpoint_iri { ?remote_obj_IRI ?pred ?obj . } .
}

```

8) How accurate and complete are the data?

relations: inputs, applies, realizes (bias):

```

SELECT ?deriv ?predicate ?args ?link ?comment
WHERE { cio:AStRIndicatorCaEU2010Ex ?predicate ?obj_property .
        ?obj_property owl:onProperty slcto:is_output . ?obj_property owl:someValuesFrom ?deriv .
        ?deriv ?p ?o . ?o owl:onProperty ?obj_props . ?o owl:someValuesFrom ?args .
        OPTIONAL { ?args cco:is_tokenized_by ?link } . OPTIONAL { ?args rdfs:comment ?comment } .
}

```

9) Are there any caveats, warnings, problems?

assumption, limitation, uncertainty:

```

SELECT ?obj_property ?keywords
WHERE { cio:AStRIndicatorCaEU2010Ex slcto:has_context ?obj_property .
        ?obj_property ?pred ?obj .
        ?obj owl:onProperty ?has_keyword . ?obj owl:someValuesFrom ?keywords .
}

```

bias

```

SELECT ?deriv ?args ?link ?comment
WHERE { cio:AStRIndicatorCaEU2010Ex ?predicate ?obj_property .
        ?obj_property owl:onProperty slcto:is_output . ?obj_property owl:someValuesFrom ?deriv .
        ?deriv ?p ?o . ?o owl:onProperty obo:BFO_0000055 . ?o owl:someValuesFrom ?args .
        OPTIONAL { ?args cco:is_tokenized_by ?link } . OPTIONAL { ?args rdfs:comment ?comment } .
}

```